/*

----------INCUBADORA------------

Incluye:

- Control de temperatura por NTC y encendido de R calefactora según TEMP

- Motor cc con control de velocidad e inversión de giro mediante L293D

- Timers: volteo, 21d de incubación y tiempo límite de motor en marcha 

- LEDs:

    verde (placa activa) - parpadea cada 0,5 segundos

    rojo (calentando)

    amarillo (21dias-pollitos)

POSIBLES MEJORAS:

- Control de giro motor. Si no se pulsa el fc, guardar sentido de giro para cambiar siguiente vez.

- Interruptor de no volteo por si se reset accidental de Arduino durante la incubación. Mejor EEPROM.Write y Read

- Separar fin de volteo de pollitosInterval para dejar de voltear a los 18 días 

- Configurar límite MAX y MIN para distintas especies

- Controlar humedad + servo leve apertura tapa

- Controlar depósito de agua

*/

//definir pines de entrada/salida y otras variables

#define ledVerde 13

#define ledRojo 12

#define ledAmbar 11

#define relePin 10

#define speedPin 9   // enable (pin 1) L293D

#define fcIzq 6

#define fcDer 7

#define motorDir1 3  // L293D Input 1 control direccion

#define motorDir2 4  // L293D Input 2 control direccion

#define ntcPin 0    // EA0 - NTC 

#define potPin 1    // EA1 - potenciómetro para controlar la elocidad de volteo

int estado=0;  //variable general para leer estados

int Mspeed = 0; // a variable to hold the current speed value 

float ntcVal=0;

float ntc_actual=0;

float ntc_anterior=0;

float temp_actual=0.0;  // int temp_actual=0;

int noVolteo=0;

int estado_LED=LOW;

long millis_volteo_anterior=0;

long millis_fc_anterior=0;

long millis_ledVerde_anterior=0;

long millis_actual = 0;

long volteoInterval = 60000;  //21600000;  // 6 horas; ahora 1 min

long fcInterval = 5000;      // 5 segundos;

long ledVerdeInterval=500;

long pollitosInterval = 300000; //1814400000; //21 dias; ahora 5 min

void setup()

{                            //inicializar variables de entrada y salida

pinMode(ledVerde, OUTPUT);

pinMode(ledRojo, OUTPUT);

pinMode(ledAmbar, OUTPUT);

pinMode(relePin, OUTPUT);

pinMode(speedPin, OUTPUT); 

pinMode(fcIzq, INPUT);

pinMode(fcDer, INPUT);

pinMode(motorDir1, OUTPUT); 

pinMode(motorDir2, OUTPUT); 

digitalWrite(motorDir1, LOW); // inicializa pines de dirección motor a paro 

digitalWrite(motorDir2, LOW);

 Serial.begin(9600);    //inicializa velocidad de comunicaciones del monitor

} 

void loop()

{

  timer();                      // controlar temporizadores

  temperatura();                // leer temperatura, presentar y actuar según valor

  delay(1000);                  // cada segundo es suficiente

}

void timer()

{

  millis_actual=millis();                                                      // lee temporizador millis

  if (millis_actual - millis_ledVerde_anterior > ledVerdeInterval)                        //cambia estado LED cada 0,5 segundos

   {

     millis_ledVerde_anterior=millis_actual;

     if (estado_LED==LOW)

       estado_LED=HIGH;

     else

       estado_LED=LOW;

     digitalWrite(ledVerde,estado_LED);

   }

  if (millis_actual - millis_volteo_anterior > volteoInterval && noVolteo==0)  // voltear si timer de volteo y no hay pollitos

   {

     millis_volteo_anterior=millis_actual;                                    

     voltear();

   }

  if (millis_actual > pollitosInterval)        // si pollitos (21d), encender LED ambar

   {

     digitalWrite(ledAmbar, HIGH);              // encender LED ambar de 21 días y poner flag de no voltear

     noVolteo=1;                              

   }

}

void temperatura()

{

  ntc_actual=analogRead(ntcPin);                //lee el valor de la NTC

  ntc_actual=ntc_anterior*0.8+ntc_actual*0.2;  //filtrado de entrada analógica

  ntc_anterior=ntc_actual;

  temp_actual=((ntc_actual-300)/8)+35.0;      //MAP no vale para float

//  temp_actual=map(ntc_actual,300,380,35.0,43.0);   // CORREGIR FORMULA (ahora solo demostración)MAP no vale para float

  Serial.print ("Temperatura = ");             // imprime valor en ºC. MEJORA: float para decimales

  Serial.print (temp_actual,1);                // imprime con un decimal;  Serial.print (temp_actual,DEC) sería sin decimales ;

  Serial.println (" grados C");

  if (digitalRead(ledRojo)==1)                // Si R calefactora encendida, comprobar temp. MAX. Real 37.8

  {

    if (temp_actual > 39.0)

    {

      digitalWrite(relePin, LOW);             // apaga R y LED rojo

      digitalWrite(ledRojo, LOW);

    } 

  }

  else

  {

    if (temp_actual < 38.0)                   // Si R apagada, comprobar límite MIN. Real 37.6

    {

      digitalWrite(relePin, HIGH);             // Si temperatura baja, enciende R y LED rojo

      digitalWrite(ledRojo, HIGH);

    } 

  }

}

void voltear()

{ 

  Mspeed = analogRead(potPin)/4;    // lee valor de Speed del potenciómetro (0-1023 a 0-255)

  analogWrite(speedPin, Mspeed);   // escribe Speed para Enable pin 1 del L293D 

  millis_actual=millis();          // leer millis

  millis_fc_anterior=millis_actual;     // inicializa timer de final de carrera

  if (digitalRead(fcDer)==1)            // Si fc derecho, motor anti-clockwise 

  {                                     // hasta que se pulse fc izquierdo o transcurra el tiempo límite fc

    while (digitalRead(fcIzq)==0 && millis_actual - millis_fc_anterior < fcInterval)      

    { 

    digitalWrite(motorDir1, HIGH);    // configurar pines de dirección del L293D para mover motor

    digitalWrite(motorDir2, LOW);

    delay (100);                      // delay para evitar rebotes

    millis_actual=millis();          // actualiza millis_actual

    }

    digitalWrite(motorDir1, LOW);   // parar motor

    digitalWrite(motorDir2, LOW);       

  } 

  else

  {                                 // si no, se supone que fc izq., mover en sentido clockwise 

    while (digitalRead(fcDer)==0 && millis_actual - millis_fc_anterior < fcInterval)   // hasta que se pulse fc derecho o transcurra el tiempo límite

    { 

    digitalWrite(motorDir1, LOW);   // configurar pines de dirección del L293D 

    digitalWrite(motorDir2, HIGH);

    delay (100);                    // delay para evitar rebotes

    millis_actual=millis();          // actualiza millis_actual

    }

    digitalWrite(motorDir1, LOW);   // parar motor 

    digitalWrite(motorDir2, LOW); 

  } 

}

